Pendulum in cartesian coordinates

RK based numerical scheme

```
> restart:
```

Pendulum equation reduced to first order mass=1

```
> EQ1  := diff(x(t),t)-u(t) ;
  EQ2  := diff(y(t),t)-v(t) ;
  EQ3  := diff(u(t),t)+x(t)*mu(t) ;
  EQ4  := diff(v(t),t)+y(t)*mu(t)+g ;
  ALG  := x(t)^2+y(t)^2-1 ;
```

$$EQ1 := \frac{d}{dt} x(t) - u(t)$$

$$EQ2 := \frac{d}{dt} y(t) - v(t)$$

$$EQ3 := \frac{d}{dt} u(t) + x(t)\,\mu(t)$$

$$EQ4 := \frac{d}{dt} v(t) + y(t)\,\mu(t) + g$$

$$ALG := x(t)^2 + y(t)^2 - 1 \qquad (1)$$

Derivate constraint 3 times

```
> DALG   := diff(ALG,t);
  DDALG  := diff(DALG,t);
  DDDALG := diff(DDALG,t);
```

$$DALG := 2\,x(t)\left(\frac{d}{dt} x(t)\right) + 2\,y(t)\left(\frac{d}{dt} y(t)\right)$$

$$DDALG := 2\left(\frac{d}{dt} x(t)\right)^2 + 2\,x(t)\left(\frac{d^2}{dt^2} x(t)\right) + 2\left(\frac{d}{dt} y(t)\right)^2 + 2\,y(t)\left(\frac{d^2}{dt^2} y(t)\right)$$

$$DDDALG := 6\left(\frac{d}{dt} x(t)\right)\left(\frac{d^2}{dt^2} x(t)\right) + 2\,x(t)\left(\frac{d^3}{dt^3} x(t)\right) + 6\left(\frac{d}{dt} y(t)\right)\left(\frac{d^2}{dt^2} y(t)\right) + 2\,y(t)\left(\frac{d^3}{dt^3} y(t)\right) \qquad (2)$$

Have a look of stable coefficients ODE

```
> expand((z+a)^3) ;
```

$$z^3 + 3\,z^2\,a + 3\,z\,a^2 + a^3 \qquad (3)$$

```
> dsolve( {diff(x(t),t,t,t)+3*a*diff(x(t),t,t)+3*a^2*diff(x(t),t)+
  a^3*x(t)} ) ;
```

$$\left\{x(t) = e^{-a\,t}\left(\_C1 + \_C2\,t + \_C3\,t^2\right)\right\} \qquad (4)$$

Stabilize constraint with Baumgarte

```
> ALGSTAB := DDDALG + 3*omega*DDALG + 3*omega^2*DALG + omega^3 *
  ALG ;
```

$$ALGSTAB := 6 \left(\frac{d}{dt} x(t)\right)\left(\frac{d^2}{dt^2} x(t)\right) + 2\, x(t) \left(\frac{d^3}{dt^3} x(t)\right) + 6 \left(\frac{d}{dt} y(t)\right)\left(\frac{d^2}{dt^2} y(t)\right) + 2\, y(t) \left(\frac{d^3}{dt^3} y(t)\right) + 3\,\omega \left(2 \left(\frac{d}{dt} x(t)\right)^2 + 2\, x(t) \left(\frac{d^2}{dt^2} x(t)\right) + 2 \left(\frac{d}{dt} y(t)\right)^2 + 2\, y(t)\left(\frac{d^2}{dt^2} y(t)\right)\right) + 3\,\omega^2 \left(2\, x(t) \left(\frac{d}{dt} x(t)\right) + 2\, y(t) \left(\frac{d}{dt} y(t)\right)\right) + \omega^3 \left(x(t)^2 + y(t)^2 - 1\right) \quad (5)$$

Solve for velocity and acceleration

```
> RESVEL  := solve( {EQ1,EQ2}, diff({x(t),y(t)},t) ) ;
  RESACC  := solve( {EQ3,EQ4}, diff({u(t),v(t)},t) ) ;
```

$$RESVEL := \left\{\frac{d}{dt} x(t) = u(t), \frac{d}{dt} y(t) = v(t)\right\}$$

$$RESACC := \left\{\frac{d}{dt} u(t) = -x(t)\, \mu(t), \frac{d}{dt} v(t) = -y(t)\, \mu(t) - g\right\} \quad (6)$$

```
> ALGSTAB := expand(subs(RESVEL,subs(RESACC, subs( RESVEL,ALGSTAB)
  ))) ;
```

$$ALGSTAB := -6\, u(t)\, x(t)\, \mu(t) - 2\, x(t) \left(\frac{d}{dt} x(t)\right) \mu(t) - 2\, x(t)^2 \left(\frac{d}{dt} \mu(t)\right) - 6\, v(t)\, y(t)\, \mu(t) - 6\, v(t)\, g - 2\, y(t) \left(\frac{d}{dt} y(t)\right) \mu(t) - 2\, y(t)^2 \left(\frac{d}{dt} \mu(t)\right) + 6\,\omega\, u(t)^2 - 6\,\omega\, x(t)^2 \mu(t) + 6\,\omega\, v(t)^2 - 6\,\omega\, y(t)^2 \mu(t) - 6\,\omega\, y(t)\, g + 6\,\omega^2 x(t)\, u(t) + 6\,\omega^2 y(t)\, v(t) + \omega^3 x(t)^2 + \omega^3 y(t)^2 - \omega^3 \quad (7)$$

Solve the stabilized equation for mu

```
> EQ5 := diff(mu(t),t) - solve( ALGSTAB, diff(mu(t),t) ) ;
```

$$EQ5 := \frac{d}{dt} \mu(t) + \frac{1}{2} \frac{1}{x(t)^2 + y(t)^2} \left(6\, u(t)\, x(t)\, \mu(t) + 2\, x(t) \left(\frac{d}{dt} x(t)\right) \mu(t) + 6\, v(t)\, y(t)\, \mu(t) + 6\, v(t)\, g + 2\, y(t) \left(\frac{d}{dt} y(t)\right) \mu(t) - 6\,\omega\, u(t)^2 + 6\,\omega\, x(t)^2 \mu(t) - 6\,\omega\, v(t)^2 + 6\,\omega\, y(t)^2 \mu(t) + 6\,\omega\, y(t)\, g - 6\,\omega^2 x(t)\, u(t) - 6\,\omega^2 y(t)\, v(t) - \omega^3 x(t)^2 - \omega^3 y(t)^2 + \omega^3\right) \quad (8)$$

Pendulum DAE reduced to index 1 and stabilized with Baumgarte

```
> EQ||(1..5) ;
```

$$\frac{d}{dt} x(t) - u(t), \frac{d}{dt} y(t) - v(t), \frac{d}{dt} u(t) + x(t)\, \mu(t), \frac{d}{dt} v(t) + y(t)\, \mu(t) + g, \frac{d}{dt} \mu(t) + \frac{1}{2} \frac{1}{x(t)^2 + y(t)^2} \left(6\, u(t)\, x(t)\, \mu(t) + 2\, x(t) \left(\frac{d}{dt} x(t)\right) \mu(t)\right. \quad (9)$$

$$+6\,v(t)\,y(t)\,\mu(t)+6\,v(t)\,g+2\,y(t)\left(\frac{d}{dt}\,y(t)\right)\mu(t)-6\,\omega\,u(t)^2+6\,\omega\,x(t)^2\,\mu(t)$$
$$-6\,\omega\,v(t)^2+6\,\omega\,y(t)^2\,\mu(t)+6\,\omega\,y(t)\,g-6\,\omega^2\,x(t)\,u(t)-6\,\omega^2\,y(t)\,v(t)$$
$$\left.-\omega^3\,x(t)^2-\omega^3\,y(t)^2+\omega^3\right)$$

Now we can use any numerical methods for ODE, for example Collatz
x'=f(x,t)

x(k+1/2) = x(k) + (DT/2) * f(x(k),t(k));
x(k+1) = x(k) + DT*f(x(k+1/2),t(k+1/2)) ;

First HALF step

```
> SUBSH := [ diff(x(t),t)   = (xH-xO)/DT,
             diff(y(t),t)   = (yH-yO)/DT,
             diff(u(t),t)   = (uH-uO)/DT,
             diff(v(t),t)   = (vH-vO)/DT,
             diff(mu(t),t)  = (muH-muO)/DT,
             x(t)           = xO,
             y(t)           = yO,
             u(t)           = uO,
             v(t)           = vO,
             mu(t)          = muO ];
```

$$SUBSH:=\left[\frac{d}{dt}\,x(t)=\frac{xH-xO}{DT},\ \frac{d}{dt}\,y(t)=\frac{yH-yO}{DT},\ \frac{d}{dt}\,u(t)=\frac{uH-uO}{DT},\ \frac{d}{dt}\,v(t)=\frac{vH-vO}{DT},\ \frac{d}{dt}\,\mu(t)=\frac{muH-muO}{DT},\ x(t)=xO,\ y(t)=yO,\ u(t)=uO,\ v(t)=vO,\ \mu(t)=muO\right] \quad \textbf{(10)}$$

```
> subs(SUBSH,[EQ||(1..5)]);
```

$$\left[\frac{xH-xO}{DT}-uO,\ \frac{yH-yO}{DT}-vO,\ \frac{uH-uO}{DT}+xO\,muO,\ \frac{vH-vO}{DT}+yO\,muO+g,\ \frac{muH-muO}{DT}+\frac{1}{2}\,\frac{1}{xO^2+yO^2}\left(6\,uO\,xO\,muO+\frac{2\,xO\,(xH-xO)\,muO}{DT}+6\,vO\,yO\,muO+6\,vO\,g+\frac{2\,yO\,(yH-yO)\,muO}{DT}-6\,\omega\,uO^2+6\,\omega\,xO^2\,muO-6\,\omega\,vO^2+6\,\omega\,yO^2\,muO+6\,\omega\,yO\,g-6\,\omega^2\,xO\,uO-6\,\omega^2\,yO\,vO-\omega^3\,xO^2-\omega^3\,yO^2+\omega^3\right)\right] \quad \textbf{(11)}$$

```
> HSTEP := op(solve( subs(SUBSH,[EQ||(1..5)]), [xH,yH,uH,vH,muH] ) ) ;
```

$$HSTEP:=\left[xH=xO+uO\,DT,\ yH=yO+vO\,DT,\ uH=uO-xO\,muO\,DT,\ vH=vO\right. \quad \textbf{(12)}$$

$$-yO\,muO\,DT-g\,DT,\ muH=-\frac{1}{2}\frac{1}{xO^2+yO^2}\big(-2\,xO^2\,muO-2\,yO^2\,muO+8\,uO\,xO\,muO\,DT+8\,vO\,yO\,muO\,DT+6\,vO\,g\,DT-6\,\omega\,uO^2\,DT+6\,\omega\,xO^2\,muO\,DT-6\,\omega\,vO^2\,DT+6\,\omega\,yO^2\,muO\,DT+6\,\omega\,yO\,g\,DT-6\,\omega^2\,xO\,uO\,DT-6\,\omega^2\,yO\,vO\,DT-\omega^3\,xO^2\,DT-\omega^3\,yO^2\,DT+\omega^3\,DT\big)\Big]$$

Second FULL step

```
> SUBSF := [ diff(x(t),t)   = (xN-xO)/DT,
             diff(y(t),t)   = (yN-yO)/DT,
             diff(u(t),t)   = (uN-uO)/DT,
             diff(v(t),t)   = (vN-vO)/DT,
             diff(mu(t),t)  = (muN-muO)/DT,
             x(t)           = xH,
             y(t)           = yH,
             u(t)           = uH,
             v(t)           = vH,
             mu(t)          = muH ];
```

$$SUBSF:=\left[\frac{d}{dt}x(t)=\frac{xN-xO}{DT},\ \frac{d}{dt}y(t)=\frac{yN-yO}{DT},\ \frac{d}{dt}u(t)=\frac{uN-uO}{DT},\ \frac{d}{dt}v(t)=\frac{vN-vO}{DT},\ \frac{d}{dt}\mu(t)=\frac{muN-muO}{DT},x(t)=xH,y(t)=yH,u(t)=uH,v(t)=vH,\mu(t)=muH\right] \quad \textbf{(13)}$$

```
> FSTEP := op(solve( subs(SUBSF,[EQ||(1..5)]), [xN,yN,uN,vN,muN]))
  ;
```

$$FSTEP:=\Big[xN=xO+uH\,DT,\ yN=yO+vH\,DT,\ uN=uO-xH\,muH\,DT,\ vN=vO-yH\,muH\,DT-g\,DT,\ muN=-\frac{1}{2}\frac{1}{xH^2+yH^2}\big(-2\,muO\,xH^2-2\,muO\,yH^2+8\,uH\,xH\,muH\,DT+8\,vH\,yH\,muH\,DT+6\,vH\,g\,DT-6\,\omega\,uH^2\,DT+6\,\omega\,xH^2\,muH\,DT-6\,\omega\,vH^2\,DT+6\,\omega\,yH^2\,muH\,DT+6\,\omega\,yH\,g\,DT-6\,\omega^2\,xH\,uH\,DT-6\,\omega^2\,yH\,vH\,DT-\omega^3\,xH^2\,DT-\omega^3\,yH^2\,DT+\omega^3\,DT\big)\Big] \quad \textbf{(14)}$$

```
> advance := proc ( x0, y0, u0, v0, mu0, dt, N )
    local kk, SUBS, SUBSBASE,
          x1, y1, u1, v1, mu1,
          xh, yh, uh, vh, muh,
          XY, UV, MU ;
    XY := [[x0,y0]] ;
    UV := [[u0,v0]] ;
    MU := [mu0] ;
    SUBSBASE := { g=9.81, DT=dt, omega=100 } ;
    for kk from 1 to N do
      # Half Step
      SUBS := SUBSBASE union { xO=XY[-1][1], yO=XY[-1][2],
                               uO=UV[-1][1], vO=UV[-1][2],
                               muO=MU[-1]} ;
      xh, yh, uh, vh, muh := op(evalf(subs( SUBS, subs( HSTEP,
```

```
[xH,yH,uH,vH,muH] )))) ;

    # Full Step
    SUBS := SUBS union { g=9.81, DT=dt, xH=xh, yH=yh,uH=uh, vH=
vh, muH=muh } ;
    x1, y1, u1, v1, mu1 := op(evalf(subs( SUBS, subs( FSTEP,
[xN,yN,uN,vN,muN] )))) ;

    XY := [op(XY),[x1,y1]] ;
    UV := [op(UV),[u1,v1]] ;
    MU := [op(MU),mu1] ;
  end  ;
  [XY,UV,MU] ;
end proc:
```

Test numerical scheme DT = 1/200

```
> RES := advance( 1, 0, 0, 0, 0, 10/2000, 2000 ) :
> plot( RES[1] ) ;
```



Test numerical scheme DT = 1/20

```
> RES := advance( 1, 0, 0, 0, 0, 10/400, 400 ) :
  plot( RES[1] ) ;
```



4 × 10^278
3 × 10^278
2 × 10^278
1 × 10^278
−2 × 10^279
−1,5 × 10^279
−10 × 10^278
−5 × 10^278
0

>

>